

R. Brigola, TH Nürnberg Georg Simon Ohm, 2014

Mathematica - Notebooks als Bonusmaterial zum Lehrbuch

**[1] Rolf Brigola Fourier-Analysis und Distributionen,
Eine Einführung mit Anwendungen,
edition swk, Hamburg 2013**

Demo-Material zum Distributionen-Kalkül

Referenzen zu Kapiteln, Abschnitten, Seitenzahlen beziehen sich im Folgenden meist auf das genannte Lehrbuch. Einige wenige weitere Referenzen findet man am Ende des Notebooks.

Die URL aller meiner *Mathematica*-Notebooks zur Fourier-Analysis ist:
<http://www.stiftung-swk.de/mathematica/>

Ziel diese Notebooks ist, Nutzern wie Teilnehmerinnen in meinen Lehrveranstaltungen einige erste Beispiele zu *Distributionen* oder wie man auch sagt zu *verallgemeinerten Funktionen* und dem Rechnen damit mit Hilfe von *Mathematica* vorzustellen.

Das Notebook ist mit *Mathematica* 10 unter Windows 7 erstellt. Dieses und weitere Demo-Notebooks findet man unter folgender URL des Autors: www.stiftung-swk.de/mathematica

1. Beispiele zu Testfunktionen

Distributionen sind stetige lineare Funktionale auf einem Vektorraum von sog. Testfunktionen. Die Stetigkeit linearer Funktionale ist durch die Topologie (d.h. den Konvergenzbegriff) im zugrunde liegenden Testfunktionenraum bestimmt. Je nach Definition eines Testfunktionenraums ergibt sich ein zugehöriger Vektorraum von Distributionen. Wir verwenden wie in [1] im Wesentlichen zwei Räume von Testfunktionen, den Raum D der unendlich oft differenzierbaren Funktionen mit kompaktem Träger und den Raum S der unendlich oft differenzierbaren schnell fallenden Funktionen. Die entsprechenden Distributionenräume werden mit D' bzw. S' bezeichnet. Je nach Beispielen im Folgenden können Testfunktionen auf \mathbb{R} oder \mathbb{R}^p definiert sein mit Werten in \mathbb{R} oder in \mathbb{C} . Man vgl. hierzu ggf. [1], Kap. 7 und 9.

Wir geben nur zwei Testfunktionen an, die für Theoriezwecke nützlich sind, an. Die Beispiele zeigen unmittelbar die Reichhaltigkeit der Räume D und S . Produkte von ϕ im ersten Beispiel mit beliebigen unendlich differenzierbaren Funktionen und Linearkombinationen davon gehören ebenfalls wieder zu D und damit auch zu S . Bzgl. Konvergenz in D und S sei auf [1], Kap. 7 und Kap. 9 verwiesen oder auf andere Literatur zur Distributionentheorie.

Beispiel 1. Die Funktion $\phi[t]=e^{-1/(1-t^2)}$ für $|t|\leq 1$, $\phi[t]=0$ sonst, gehört zu D. Damit z.B. auch jedes Produkt $f\phi$ mit irgendeiner unendlich oft differenzierbaren Funktion f . Wir betrachten den Graphen:

In[1]=


```
ClearAll["Global`*"]; Remove["Global`*"]; ? Global`*
```

```
Remove::rmnsm : There are no symbols matching "Global`*". >>
```

```
Information::nomatch : No symbol matching Global`* found. >>
```

```
In[2]=  $\phi[t_] := \text{Piecewise}[\{\{e^{-1/(1-t^2)}, -1 < t < 1\}\}, 0]$ 
Plot[ $\phi[t]$ , {t, -2, 2}, PlotStyle  $\rightarrow$  Directive[
  Blue, Thickness[0.006]], PlotRange  $\rightarrow$  All]
```


Out[3]=



Beispiel 2. Die Gaußfunktion $\text{gauss}[f] = e^{-t^2/2}$ für reelle t gehört zu S. Nachfolgend ihr Graph.

```
In[4]=  $\text{gauss}[t_] := \text{Exp}[-t^2/2];$ 
Plot[ $\text{gauss}[t]$ , {t, -4, 4}, PlotStyle  $\rightarrow$  Directive[
  Blue, Thickness[0.006]], PlotRange  $\rightarrow$  All]
```

Out[5]=



2. Beispiele von Distributionen

Die für Themen meiner *Mathematica*-Notebook-Serie - die vor allem für Zwecke in der Lehre für Ingenieurstudenten dienen soll - wichtigsten Distributionen sind die Diracsche Delta-Distribution, lokal-integrierbare Funktionen und rationale Funktionen. Wir betrachten im Folgenden Beispiele dazu.

Ich notiere die Anwendung einer Distribution T auf eine Testfunktion ϕ wie weithin üblich durch $\langle T, \phi \rangle$ oder, um einen Parameter aus dem zugrundeliegenden Parameterraum wie etwa einen Zeitparameter t oder einen Kreisfrequenzparameter ω mit anzugeben, durch $\langle T(t), \phi(t) \rangle$ oder $\langle T(\omega), \phi(\omega) \rangle$. Bei Physikern und Ingenieuren ist auch üblich, stattdessen die Notation $\int T(t) \phi(t) dt$ zu verwenden. Sie ist für lokal-integrierbare Funktionen T , die man als reguläre Distributionen auffasst, sinnvoll. In Fällen, bei denen T eine singuläre Distribution ist, ist *das Integral nur eine formale Schreibweise*, die auf die Eigenschaften des Integrals als Linearform Bezug nimmt. Auch in *Mathematica* findet diese Notation Verwendung, wie wir sehen werden.

2.1 Die Dirac-Distribution, auch Dirac-Impuls oder wie in *Mathematica* DiracDelta genannt.

Man kann die Dirac-Distribution δ charakterisieren als **Abtastfunktional** an der Stelle $t=0$ für Testfunktionen aus D oder aus S , d.h. für ϕ aus D oder aus S gilt

$$\langle \delta, \phi \rangle = \phi(0)$$

Das Abtastfunktional für eine Stelle $t = t_0$ schreiben wir als $\delta(t - t_0)$, also

$$\langle \delta(t - t_0), \phi(t) \rangle = \phi(t_0).$$

Zu anderen Charakterisierungen, etwa δ als verallgemeinerte Ableitung der Einheitssprungfunktion, als Faltungseins, als Impulsmodell in Physik und Elektrotechnik oder als Punktmasse in der Maßtheorie kommen wir etwas später noch. In *Mathematica* wird δ mit **DiracDelta** bezeichnet. Hier nun die Abtasteigenschaft mit *Mathematica*:

```
In[7]:= Integrate[DiracDelta[t] gauss[t], {t, -∞, +∞}]
```

```
Out[7]= 1
```

```
In[8]:= Integrate[DiracDelta[t - 1/2] phi[t], {t, -1, 1}]
```

```
Out[8]= 1/e^{4/3}
```

2.2 Lokal-integrierbare Funktionen, verallgemeinerte Ableitung

Jede lokal-integrierbare Funktion f lässt sich als Distribution auffassen, d.h. als stetiges lineares Funktional auf D vermöge der Zuordnung

$$\langle f, \phi \rangle = \int f(t) \phi(t) dt.$$

Dabei verwendet man den Lebesgueschen Integralbegriff. Alle solchen Distributionen nennt man **reguläre Distributionen**. Mit dieser Definition und bekannter Integralrechnung stehen bereits sehr viele Distributionen zur Verfügung und auch eine ganze Reihe von Eigenschaften, die aus den Eigenschaften von Integralen herzuleiten sind.

Eine der wichtigsten Eigenschaften folgt für *differenzierbare* f mit **partieller Integration** und der Voraussetzung, dass Testfunktionen ϕ aus D kompakte Träger haben. Für ϕ aus D gilt dann

$$\langle f', \phi \rangle = \int f'(t) \phi(t) dt = - \int f(t) \phi'(t) dt = - \langle f, \phi' \rangle.$$

In Verallgemeinerung definiert man für beliebige Distributionen T die sog. **verallgemeinerte Ableitung**, hier ebenfalls notiert als T' , durch

$$\langle T', \phi \rangle = - \langle T, \phi' \rangle \quad \text{und} \quad \langle T^{(n)}, \phi \rangle = (-1)^n \langle T, \phi^{(n)} \rangle \quad \text{für die } n\text{-te verallgemeinerte Ableitung.}$$

Da alle Testfunktionen als unendlich oft differenzierbar vorausgesetzt sind, ist damit jede Distribution mit der verallgemeinerten Ableitung beliebig oft differenzierbar. Diese Eigenschaft hat es etwa Mitte des letzten Jahrhunderts, seitdem die Theorie entwickelt wurde, ermöglicht *Differentialgleichungen als Gleichungen zwischen Distributionen zu lesen* und den Lösungsbegriff dafür zu erweitern. Erst dies hat z.B. eine vollständige Lösungstheorie für lineare Differentialgleichungen mit konstanten Koeffizienten ermöglicht.

Beispiel. Zu Ableitungen der Dirac-Distribution: $(t-1/2) \delta'''[t-1/2] = -3 \delta''[t-1/2]$.

(Empfehlung: Erstmal per Hand lösen). Im dritten Beispiel sieht man: $t \delta'(t) = -\delta(t)$, allgemein gilt $t \delta^{(m)} = -m \delta^{(m-1)}$. **Translationen und Ableitungen sind vertauschbar.**

```
In[92]:= Integrate[(t - 1/2) DiracDelta'''[t - 1/2] gauss[t], {t, -∞, +∞}]
```

```
Out[92]= 9 / (4 e^(1/8))
```

```
In[90]:= Integrate[-3 DiracDelta''[t - 1/2] gauss[t], {t, -∞, +∞}]
```

(* Produktregel beim Differenzieren *)

```
Out[90]= 9 / (4 e^(1/8))
```

```
In[11]:= Integrate[t D[DiracDelta[t], t] gauss[t], {t, -∞, +∞}]
```

```
Out[11]= -1
```

Wichtige Erkenntnis: **Die Dirac-Distribution ist die verallgemeinerte Ableitung der Einheitssprungfunktion**, letztere in *Mathematica* als HeavisideTheta implementiert.

Da am Ausgang von Differentiatoren in der Elektrotechnik bei der Aufschaltung von Gleichspannung - im mathematischen Modell also Input U_0 HeavisideTheta[t] - Spannungsimpulse erscheinen, wird die **Dirac-Distribution auch als Dirac-Impuls** bezeichnet.

Allgemein ergibt verallgemeinerte Ableitung an Sprungstellen t_0 ansonsten differenzierbarer Funktionen f Dirac-Impulse mit Impulsstärke $(f[t_{0+}] - f[t_{0-}])$.
Hier zwei Beispiele.

```
In[12]:= D[UnitStep[t], t] (* Ableitung der Einheitssprungsfunktion
      UnitStep zeigt nur den klassischen Ableitungsanteil *)
```


```
Out[12]= { Indeterminate t == 0
          { 0             True
```

```
In[13]:= D[HeavisideTheta[t], t] (* Ableitung von HeavisideTheta wird in Mathematica
      nach Distributionenkalkül im verallgemeinerten Sinn ausgeführt *)
```

```
Out[13]= DiracDelta[t]
```

```
In[14]:= f[t_] := t (HeavisideTheta[t] - HeavisideTheta[t - 1]) +
      3 HeavisideTheta[t - 1] (* Eine Funktion f mit Knick- und Sprungstelle,
      abschnittsweise mit HeavisideTheta (quasi als "Schneidmesser" benutzt)
      definiert, ergibt die verallgemeinerte Ableitung von f *)
```

```
In[15]:= Plot[f[t], {t, -1, 2}, PlotStyle -> Directive[
      Blue, Thickness[0.006]], PlotRange -> All]
```




```
In[16]:= FullSimplify[D[f[t], t]] (* Ableitung von f;
      das Ergebnis in noch nicht sehr übersichtlicher Form *)
```

```
Out[16]= 3 DiracDelta[-1 + t] + t (-DiracDelta[-1 + t] + DiracDelta[t]) -
      HeavisideTheta[-1 + t] + HeavisideTheta[t]
```

Der Plot zeigt nur den klassischen Anteil der verallgemeinerten Ableitung, eine Distribution wie DiracDelta wird dabei nicht dargestellt.

```
In[17]:= Plot[3 DiracDelta[-1 + t] + t (-DiracDelta[-1 + t] + DiracDelta[t]) -
  HeavisideTheta[-1 + t] + HeavisideTheta[t], {t, -2, 2}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
```

Out[17]=



Hier nochmals das Ergebnis für $D[f[t], t]$ in übersichtlicherer Form, wenn wir als *Zusatzwissen aus dem Distributionenkalkül* verwenden, dass $t \delta(t) = 0$ ist und $t \delta(t-1) = \delta(t-1)$. Dies wird im nächsten Kommando mit /. usw. mitgeteilt. Allgemein gilt für unendlich oft differenzierbare f :

$$f(t) \delta(t - t_0) = f(t_0) \delta(t - t_0).$$

Nachfolgend der klassische Anteil der Ableitung + ein Impuls der Stärke 2 an der Stelle $t_0=1$.

```
In[18]:= D[f[t], t] /. t (-DiracDelta[-1 + t] + DiracDelta[t]) -> -DiracDelta[-1 + t]
```

```
Out[18]= 2 DiracDelta[-1 + t] - HeavisideTheta[-1 + t] + HeavisideTheta[t]
```

2.3 Cauchy-Hauptwert, Pseudofunktionen, rationale Funktionen als Pseudofunktionen

Eine für Anwendungen besonders in der Theorie zeitinvarianter linearer Systeme wichtige Klasse von Funktionen sind rationale Funktionen, die sich - wie im Folgenden demonstriert - in geeigneter Weise ebenfalls als Distributionen auffassen lassen.

Die Funktion $f(t)=1/t$ ist nicht lokal-integrierbar und damit nicht unmittelbar als Distribution aufzufassen. Es existiert jedoch für jedes ϕ aus D

$$\langle \text{vp}(f), \phi \rangle = \lim_{\epsilon \rightarrow 0^+} \left(\int_{-\infty}^{\epsilon} f(t) \phi(t) dt + \int_{\epsilon}^{+\infty} f(t) \phi(t) dt \right) \quad (\text{vgl. [1], S. 134}).$$

Diese Distribution $\text{vp}(f)$ nennt man eine **Regularisierung von f** und heißt **Cauchy-Hauptwert** von $f(t)=1/t$ (von frz. "valeur principal") oder sie wird auch als **Pseudofunktion** $\text{pf}(t^{-1})$ notiert (von frz. "partie finie" nach M. Hadamard). Ihre verallgemeinerten Ableitungen erzeugen weitere Pseudofunktionen $\text{pf}(t^{-n})$, $n \in \mathbb{N}$. Der Cauchy-Hauptwert $\text{pf}(t^{-1}) = \text{vp}(1/t)$ ist die **verallgemeinerte Ableitung der regulären Distribution** $\ln(|t|)$ aus S' , gehört daher wie die anderen noch angesprochenen Pseudofunktionen ebenfalls zu S' . Für Testfunktionen ϕ , deren Träger Null nicht enthält ist $\langle \text{vp}(f), \phi \rangle$ einfach das konvergente Integral über \mathbb{R}

$$\langle \text{vp}(f), \phi \rangle = \int_{-\infty}^{+\infty} \frac{\phi(t)}{t} dt = \int_0^{+\infty} \frac{\phi(t) - \phi(-t)}{t} dt.$$

Auf analoge Weise kann man auch für andere Funktionen mit algebraischen Singularitäten wie etwa $\tan(t)$, $\cot(t)$ oder $1/\sin(t)^2$ Regularisierungen finden. Diese sind dann wegen der Periodizität der genannten Beispiele **periodische Distributionen** und besitzen **Darstellungen als verallgemeinerte Fourierreihen**. Etwas mehr dazu weiter unten in 3.4. Leser, die das Thema vertiefen möchten, seien neben [1] auf die am Ende genannten Quellen [2]-[6] hingewiesen.

Als Folge unseres Beispiels *halten wir fest, dass alle rationalen Funktionen der Form $1/(\pm i\omega - \pm b)^k$ und $1/(\pm i\omega - (r \pm i b))^k$ für $b \in \mathbb{R}$ und $r \in \mathbb{R} \setminus \{0\}$ als \mathbb{C} -wertige Distributionen mit reellem Parameter ω zu \mathcal{S}' gehören*. Sie spielen in der gesamten Kommunikationstechnik eine bedeutende Rolle, denn sie treten - wie wir in einem späteren Notebook dieser Reihe noch sehen werden - als **Fouriertransformierte typischer Lösungsanteile von gewöhnlichen linearen Differentialgleichungen mit konstanten Koeffizienten** auf (mit der Kreisfrequenz ω als Parameter) auf. Mit solchen Differentialgleichungen können sehr viele zeitinvariante lineare Übertragungssysteme (auch als LTI-Systeme oder LZI-Systeme bezeichnet) mathematisch beschrieben werden.

Demo-Beispiele:

In[19]:= `D[Log[Abs[t]], t] (* zeigt 1/t als Ableitung von ln(|t|) *)`

Out[19]=
$$\frac{\text{Abs}'[t]}{\text{Abs}[t]}$$

In[20]:= `D[Log[Abs[t]], {t, 2}] (* -1/t^2 als 2. Ableitung von ln(|t|) *)`

Out[20]=
$$-\frac{\text{Abs}'[t]^2}{\text{Abs}[t]^2} + \frac{\text{Abs}''[t]}{\text{Abs}[t]}$$

In[21]:= `Integrate[1/t gauss[t], {t, -∞, ∞}, PrincipalValue -> True]`
 (* $\langle \text{vp}(t^{-1}), \text{gauss} \rangle = 0$ *)

Out[21]= 0

In[22]:= `Integrate[1/t gauss[t], {t, -∞, ∞}]`
 (* $\text{gauss}[t]/t$ ist nicht integrierbar *)

Integrate::div : Integral of $\frac{e^{-\frac{t^2}{2}}}{t}$ does not converge on $\{-\infty, \infty\}$. >>

Out[22]=
$$\int_{-\infty}^{\infty} \frac{e^{-\frac{t^2}{2}}}{t} dt$$

3. Beispiele zum Rechnen mit Distributionen

3.1 Zu Gleichungen zwischen Distributionen

Gleichungen zwischen Distributionen können mehr Lösungen besitzen als dieselben Gleichungen zwischen klassischen Funktionen.

Wir betrachten als Beispiele die allgemeinen Lösungen der Gleichungen

$$(1) \quad t^n T(t) = 0 \text{ und}$$

$$(2) \quad t^n T(t) = 1$$

für gesuchte Distributionen T .

Zu beachten ist zunächst, dass aus $t \delta(t) = 0$ für $0 \leq k < n$ schnell folgt

$$(a) \quad t \delta^{(k)}(t) = -k \delta^{(k-1)} \quad (\text{Übung: Verwenden Sie vollständige Induktion})$$

$$(b) \quad t^n \delta^{(k)}(t) = 0 \quad (\text{Übung: folgt aus (a)})$$

Als Folge hat die Gleichung (1) die Lösung $T_A = \sum_{k=0}^{n-1} c_k \delta^{(k)}$ mit beliebigen Konstanten c_k eine Lösung von (1). Dass es keine weiteren Lösungen gibt ist in [1], S. 143 gezeigt, d.h. T_A ist die allgemeine Lösung der homogenen Gleichung (1) in D' oder S' . **Die Gleichung (2) hat daher die allgemeine Lösung $T(t) = \text{pf}(t^{-n}) + T_A$.**

Wir verwenden *Mathematica* zur Demonstration und betrachten die homogene Differentialgleichung

$$(3) \quad t T'(t) + 3 T(t) = 0.$$

`DSolve` ergibt die klassischen Lösungen c/t^3 (c beliebige Konstante). Als Gleichung für Distributionen T ergeben sich wegen der Beziehung (a) auch noch die weiteren Lösungen $T(t) = k \delta^{(2)}(t)$, k beliebige Konstante. (Allgemeiner hat $t T(t) + m T(t) = 0$ die Distributionen $k \delta^{(m-1)}$ als Lösungen).
Hier zur Ansicht:

```
In[23]:= DSolve[t T'[t] + 3 T[t] == 0, T[t], t]
```

```
(* DSolve ergibt nur die klassischen Lösungen c/t^3 *)
```

```
Out[23]= {{T[t] -> C[1]/t^3}}
```

Nun die Verifikation der obigen Feststellung mit *Mathematica* durch Einsetzen von $T(t) = C/t^3 + K \delta^{(2)}(t)$ in die Gleichung (3). Hier verwendet

Mathematica die obige Gleichung (a):

```
In[24]:= Simplify[t C D[1/t^3, t] + 3 C/t^3 +
  (t K D[DiracDelta[t], {t, 3}] + 3 K D[DiracDelta[t], {t, 2}])]
```

Out[24]= 0

3.2 Zur Konvergenz von Distributionenfolgen

Eine Folge von Distributionen T_n aus D' bzw. S' konvergiert in D' bzw. S' gegen eine Distribution T , wenn für alle Testfunktionen ϕ aus D bzw. S gilt

$$\langle T, \phi \rangle = \lim_{n \rightarrow \infty} \langle T_n, \phi \rangle.$$

Wir betrachten einige Beispiele.

Beispiel 1. Für irgendeine integrierbare Funktion f mit $\int f(t) dt = 1$ konvergiert die Folge $f_n(t) = nf(nt)$ - die f_n aufgefasst als reguläre Distributionen in D' oder S' - distributionell gegen die Dirac-Distribution. Wir wählen für unser Beispiel etwa die Funktion $f(t) = 1/(\pi(1+t^2))$, wenden f_n auf die Gaussfunktion an und bilden den Limes für $n \rightarrow \infty$. Das Ergebnis ist $1 = \text{gauss}(0) = \langle \delta, \text{gauss} \rangle$.

```
f[n_, t_] = n / (pi (1 + (n t)^2))
Limit[Integrate[f[n, t] gauss[t], {t, -Infinity, +Infinity}], n -> Infinity]
```

Out[25]=
$$\frac{n}{\pi (1 + n^2 t^2)}$$

Out[26]= 1

Eine einfache Limesbildung der Folge $f[n,t]$ für $n \rightarrow \infty$ mit *Mathematica* ergibt dagegen Null als punktwisen Limes der Funktionenfolge für alle $t \neq 0$ und lässt den Nullpunkt, der gerade der Träger von δ ist, unberücksichtigt.

```
In[27]:= Limit[f[n, t], n -> Infinity]
```

Out[27]= 0

Nachfolgend ein Plot von $f[100,t]$, der "eine Nadel um Null" als reguläre Näherung für δ zeigt und der Wert $\langle f[20,t], \text{gauss}[t] \rangle$ als Näherung für $\langle \delta, \text{gauss} \rangle$.


Das Bild entspricht in etwa der vielfach gebrauchten Vorstellung, dass ein Dirac-Impuls "so etwas wie eine Funktion sei, die überall Null wird, im Nullpunkt aber gegen unendlich strebt". Richtig ist, dass man mit Funktionen wie $f[n,t]$ für große n klassische Funktionen als reguläre Näherungen für δ erhält, sodass $\langle f[n,t], \phi[t] \rangle \approx \langle \delta, \phi \rangle$. Dass diese Vorstellung nicht unbedingt den ganzen Sachverhalt beschreibt, sehen wir in den nachfolgenden Beispielen 2 und 3.

```

NIntegrate[f[20, t] gauss[t], {t, -Infinity, + Infinity}]
(* Näherung für  $\delta$ , gauss  $\approx 1$  *)
Plot[f[100, t], {t, -1, 1}, PlotRange -> {{-1, 1}, {0, 35}}, Frame -> True,
  FrameStyle -> Directive[Black, FontSize -> 22, FontWeight -> Plain],
  PlotStyle -> {Blue, Thickness[0.006]}]

```

Out[28]= 0.961323



Beispiel 2. Hier als Beispiel die Funktionenfolge $\text{sinc}(nt) = \sin(nt)/(\pi t)$ nicht absolut-integrierbarer Funktionen, die in der Kommunikationstechnik eine herausragende Rolle spielen. Diese Funktionen verschwinden auch außerhalb des Nullpunkts nie, sondern oszillieren immer stärker, konvergieren aber distributionell ebenfalls gegen δ , da die immer stärkeren Oszillationen für $n \rightarrow \infty$ in Integralen über Intervalle, die den Nullpunkt nicht enthalten, zu Auslöschung führen. Wir testen wieder die Näherung $\langle \text{sinc}(20t), \text{gauss}(t) \rangle$ und vergleichen mit der Näherung oben mittels $f(20,t)$. Obwohl $\text{sinc}(20t)$ "keine Nadelgestalt" hat, ist die Näherung für $\langle \delta, \text{gauss} \rangle$ besser als oben, mit der verwendeten Rechengenauigkeit sogar im Ergebnis exakt.


```

In[30]:= sinc[n_, t_] = Sin[n t] / (π t)
NIntegrate[sinc[20, t] gauss[t], {t, -Infinity, +Infinity}]
(*Näherung für <δ, gauss>*)
Plot[sinc[50, t], {t, -1, 1}, PlotRange → {{-1, 1}, {-4, 20}}, Frame → True,
FrameStyle → Directive[Black, FontSize → 22, FontWeight → Plain],
PlotStyle → {Blue, Thickness[0.006]}]

```

Out[30]= $\frac{\text{Sin}[n t]}{\pi t}$

Out[31]= 1.



Beispiel 3. Garnichts mehr gemeinsam mit einer Nadelgestalt haben auch die Funktionen $g(n,t)=n \sin(nt) \text{ HeavisideTheta}(t)$. Auch diese Folge konvergiert distributionell gegen δ und ergibt mit $\langle g(20, t), \text{gauss}(t) \rangle$ immer noch eine bessere Näherung für $\langle \delta, \text{gauss} \rangle$ als "die Nadel" bei gleichem n im Beispiel 1. δ ist also auch D'-Limes dieser Folge.


```

g[n_, t_] = n Sin[n t] HeavisideTheta[t]
NIntegrate[g[20, t] gauss[t], {t, 0, +Infinity}] (*Näherung für<math>\delta, \text{gauss}>*)
Plot[g[40, t], {t, -1, 1}, PlotRange -> {{-1, 1}, {-42, 42}}, Frame -> True,
  FrameStyle -> Directive[Black, FontSize -> 22, FontWeight -> Plain],
  PlotStyle -> {Blue, Thickness[0.006]}]

```

Out[33]= n HeavisideTheta[t] Sin[n t]

Out[34]= 1.00252



Eine wichtige Erkenntnis ist, dass jede distributionell konvergente Reihe $\sum_{k=-\infty}^{+\infty} T_k = T$ von Distributionen T, T_k anders als im klassischen Fall gliedweise (verallgemeinert) differenziert werden kann und dass dann gilt $\sum_{k=-\infty}^{+\infty} T_k' = T'$. Wir betrachten zwei Beispiele.

Erste verallgemeinerte Fourierreihen

Beispiel 4. Verallgemeinerte Fourierreihe einer Impulsfolge

Die Reihe $\sum_{k=1}^{+\infty} \sin(kt)/k$ stellt die Sägezahnfunktion $f(t)=(\pi-t)/2$ für $0 < t < 2\pi$, $f(0)=0$ dar. Sie kann als reguläre 2π -periodische Distribution aufgefasst werden. Ihre gliedweise Ableitung ist im klassischen Sinn überall divergent. **Als Distributionenreihe konvergiert aber $\sum_{k=1}^{+\infty} \cos(kt)$ gegen $T(t) = -1/2 + \pi \sum_{k=-\infty}^{+\infty} \delta(t - 2\pi k)$** (vgl. [1], S. 170). Anders gesagt: Die **Dirichletkerne** $1 + 2 \sum_{k=-\infty}^{+\infty} \cos(kt)$ (vgl. [1], S.12) konvergieren distributionell gegen die 2π -periodische Impulsfolge $2\pi \sum_{k=-\infty}^{+\infty} \delta(t - 2\pi k) = \sum_{k=-\infty}^{+\infty} e^{ikt}$. Damit haben wir ein erstes Beispiel einer sog. **verallgemeinerten Fourierreihe**. Wir testen als Beispiel für $n=5, 30$ und 55

a) $\langle 1 + 2 \sum_{k=1}^n \cos(kt), \phi_1(t) \rangle$ als Näherungen für

b) $\langle 2\pi \sum_{k=-1}^{+1} \delta(t - 2k\pi), \phi_1(t) \rangle$

wobei $\phi_1(t) = \phi(t/(3\pi))$ sei, mit der Testfunktion ϕ aus Beispiel 1 ganz oben.

Es genügt daher über den Träger $[-3\pi, 3\pi]$ von ϕ_1 zu integrieren. Für b) setzen wir direkt die auftretenden 3 Abtastwerte ein und erkennen die Verbesserungen der Näherungen durch Hinzunahme von mehr Cos-Schwingungen. *Die Partialsummen der verwendeten Reihe sind also wieder klassische, hier sogar unendlich oft differenzierbare Funktionen als Näherungen für eine distributionelle Impulsfolge.*

b) ergibt wegen der beschränkten Träger von Testfunktionen ϕ_1 aus D de facto immer eine endliche Summe von Abtastwerten, bei ϕ_1 aus S eine absolut-konvergente Reihe von Abtastwerten.

```
In[36]:= Table[NIntegrate[(1 + 2 Sum[Cos[k t], {k, 1, n}]) phi[t/(3 pi)], {t, -3 pi, 3 pi},
  {n, 5, 55, 25}]
```

```
N[2 pi Sum[phi[2 k pi/(3 pi)], {k, -1, 1}]]
```

```
Out[36]= {4.38917, 4.38866, 4.38866}
```

```
Out[37]= 4.38866
```

Fazit: Wir halten fest, dass alle Fourierreihen, deren **Koeffizienten c_k polynomial beschränkt sind** (d.h. für alle c_k ($k \in \mathbb{Z} \setminus \{0\}$) gilt $|c_k| \leq C |k|^m$ mit geeigneten Konstanten C und $m \in \mathbb{N}$) **distributionell konvergieren** (vgl. [1], Abschnitt 8.1). Man spricht dann von **verallgemeinerten Fourierreihen**.

Beispiel 5. Verallgemeinerte Fourierreihe des Cotangens als Beispiel zur Regularisierung einer Funktion mit einer algebraischen Singularität

Die Cotangens-Funktion ist nicht lokal-integrierbar und besitzt keine klassisch konvergente Fourierreihendarstellung. Jedoch ergibt die (klassische) Fourierreihe von

$$T(t) = \ln|2 \sin(t)| \text{ für } t \neq (2k+1)\pi/2$$

(vgl. [1], S. 175) durch gliedweise Differentiation die verallgemeinerte Fourierreihe

$$T'(t) = 2 \sum_{k=1}^{\infty} \sin(2kt).$$

Diese Reihe ist eine Regularisierung \cot_r des Cotangens, d.h.

$$\langle T'(t), \phi \rangle = - \langle T, \phi' \rangle = \langle \cot_r(t), \phi(t) \rangle \text{ für } \phi \text{ aus D oder S.}$$

Verallgemeinerte Fourierreihen wie T' treten also typischerweise als Regularisierungen periodischer Funktionen mit Singularitäten wie beim Cotangens auf. Sie sind immer eine geeignet hohe verallgemeinerte Ableitung einer stetigen periodischen Funktion.

Zunächst zu den Fourierreihenentwicklungen von T und T' :


In[38]:= **T[t_] = Log[Abs[2 Sin[t]]]**

Out[38]= Log[2 Abs[Sin[t]]]

In[39]:= **FourierCosSeries[T[t], t, 12] (* Eine Partialsumme der FR-Entwicklung von T *)**


Out[39]= $-\cos[2t] - \frac{1}{2}\cos[4t] - \frac{1}{3}\cos[6t] - \frac{1}{4}\cos[8t] - \frac{1}{5}\cos[10t] - \frac{1}{6}\cos[12t]$

In[40]:= **Plot[-Cos[2 t] - $\frac{1}{2}$ Cos[4 t] - $\frac{1}{3}$ Cos[6 t] - $\frac{1}{4}$ Cos[8 t] - $\frac{1}{5}$ Cos[10 t] - $\frac{1}{6}$ Cos[12 t], {t, - π , π }, PlotStyle -> {Blue, Thickness[0.006]}]**



In[41]:= **Plot[T[t], {t, - π , π }, PlotStyle -> {Blue, Thickness[0.006]}]**


(* hier die Funktion T(t) in $[-\pi, \pi]$ *)



In[42]:= **FullSimplify[D[T[t], t]] (* Ihre Ableitung ergibt den Cotangens *)**

Out[42]= Cot[t] Sign[Sin[t]] Abs'[Sin[t]]

```
In[43]:= Plot[Cot[t], {t, -π, π}, PlotRange → Automatic,
  PlotStyle → {Blue, Thickness[0.006]}]
```



Der Cotangens hat keine klassische Fourierreihendarstellung, *Mathematica* gibt daher auch kein Ergebnis zurück:

```
In[44]:= FourierSeries[Cot[t], t, 12]
```

```
Out[44]= FourierSeries[Cot[t], t, 12]
```

Hier nun die verallgemeinerte Fourierreihe einer Regularisierung \cot_r des Cotangens.

Man erkennt im Bild unten je nach Anzahl der Summanden in der verwendeten Partialsumme wachsend hochfrequente Oszillationen, die außerhalb einer Nullumgebung dem Verlauf der Cotangensfunktion folgen und um Null herum die Singularität des Cotangens dadurch "beheben", dass sie als C^∞ -Funktionen durch Null laufen. Die verallgemeinerte Fourierreihe konvergiert klassisch nirgendwo in $]0, \pi[$ ($\pi/2$) (Übung: Beweisen Sie das). Mit zunehmendem Grad der trigonometrischen Näherungspolynome rückt die "Singularitäten-Behebung" näher an Null heran. Im Ergebnis erhält man mit der kompletten verallgemeinerten Fourierreihe eine Regularisierung des Cotangens wie mit Hilfe des Cauchy-Hauptwerts $\text{vp}(t^{-1})$, indem man für Testfunktionen ϕ mit Träger in einer kleinen Nullumgebung den $\cot(t)$ durch $t \cot(t) \text{vp}(t^{-1})$ ersetzt, d.h.

$$(\#) \quad \langle \cot_r, \phi \rangle = \langle \text{vp}(t^{-1}), t \cot(t) \phi(t) \rangle = \int_0^\infty (\phi(t) - \phi(-t)) \cot(t) dt$$

Mehr Details über Regularisierungen findet man bei Interesse in [2], Ch. I, 3.7.


Wir definieren $\cot_r[n,t]$ nachfolgend als Partialsumme der verallgemeinerten Fourierreihe von T^{-1} mit n Summanden und betrachten diese zusammen mit dem Verlauf des Cotangens

- mit 20 Summanden, aus Symmetriegründen nur im Intervall $[0, \pi/2]$
- mit 50 Summanden von $-\pi$ bis π

```

In[45]:= cot_r [n_, t_] := 2 Sum[ Sin[2 k t], {k, 1, n}]
p1 := Plot[cot_r[20, t], {t, 0,  $\pi/2$ },
  PlotStyle -> {Red, Thickness[0.0036]}, PlotRange -> {-5, 30}]
p2 := Plot[Cot[t], {t, 0,  $\pi/2$ }, PlotStyle -> {Blue, Thickness[0.0036]},
  PlotRange -> {-5, 30}]
Show[{p1, p2}]


```



```

In[49]:= p3 := Plot[cot_r[50, t], {t, - $\pi/2$ ,  $\pi/2$ },
  PlotStyle -> {Red, Thickness[0.0036]}, PlotRange -> {-80, 80}]
p4 := Plot[Cot[t], {t, - $\pi/2$ ,  $\pi/2$ }, PlotStyle -> {Blue, Thickness[0.0036]},
  PlotRange -> {-80, 80}] (* Cot in Blue *)
Show[
  {p3,
   p4}]


```



Zum Testen der Regularisierung cot_r wählen wir ϕ_2 und ϕ_3 als skalierte Translationen der ganz oben schon eingeführten Testfunktion ϕ und betrachten Näherungswerte für $\langle T', \phi_2 \rangle$ und für $\langle T', \phi_3 \rangle$, die wir durch numerische Integration erhalten:

```
In[52]:=  $\phi_2[t_] = 5 \phi[2(t - 0.25)]$ 
Plot[ $\phi_2[t]$ , {t, -.5, 1}, PlotStyle → Directive[
  Blue, Thickness[0.006]], PlotRange → All]
```

```
Out[52]=  $5 \left( \begin{cases} e^{-\frac{1}{1-4(-0.25+t)^2}} & -1 < 2(-0.25+t) < 1 \\ 0 & \text{True} \end{cases} \right)$ 
```



```
In[54]:= NIntegrate[Cotr[200, t]  $\phi_2[t]$ , {t, -0.25, 1}, WorkingPrecision → 40]
(* Partialsumme von Cotr angewandt auf  $\phi_2$  *)
```

```
Out[54]= 3.639261611374102827614682534232306522817
```

```
In[55]:= NIntegrate[-T[t]  $\phi_2'[t]$ , {t, -0.25, 1}, WorkingPrecision → 20]
(* hier Gleiches mit partieller Integration unter Verwendung von T(t) *)
```

```
Out[55]= 3.6392617287678451920
```

```
In[56]:= NIntegrate[Cot[t] ( $\phi_2[t] - \phi_2[-t]$ ), {t, 0, 1}, WorkingPrecision → 20]
(* hier Gleiches mit der Regularisierung des Cotangens wie (#) oben *)
```


```
Out[56]= 3.6392617287678464743
```

Die unterschiedlichen Ergebnisse bei der numerischen Integration weisen auf die Problematik bei der Integration stark oszillierender Integranden hin. Details und Referenzen zu diesem Thema der Numerischen Mathematik finden interessierte Leser etwa in der unten genannten Arbeit [7] von *E. Novak, M. Ullrich, H. Wozniakowsky*.

Hier ein weiterer Test mit ϕ_3 wie nachfolgend definiert:


```
In[57]:=  $\phi_3[t_] = 5 \phi[5t - 1.5]$ 
Plot[ $\phi_3[t]$ , {t, -.5, 1}, PlotStyle -> Directive[
Blue, Thickness[0.006]], PlotRange -> All]
```

$$\text{Out[57]= } 5 \left(\begin{cases} e^{-\frac{1}{1 - (-1.5 + 5t)^2}} & -1 < -1.5 + 5t < 1 \\ 0 & \text{True} \end{cases} \right)$$



```
In[59]:= f[t_] =  $\phi_3'[t]$  (* f die Ableitung von  $\phi_3$  *)
Plot[f[t], {t, 0.1, 0.5}, PlotStyle -> Directive[
Blue, Thickness[0.006]], PlotRange -> All]
```

$$\text{Out[59]= } 5 \left(\begin{array}{l} 0 \\ - \left(\left(10 \cdot e^{-\frac{1}{1 - (-1.5 + 5t)^2}} \cdot (-1.5 + 5t) \right) / \left(1 - (-1.5 + 5t)^2 \right)^2 \right) \\ 0 \\ 0 \end{array} \right) \begin{array}{l} t < \frac{1}{10} \mid \mid t = \frac{1}{10} \\ \frac{1}{10} < t < \frac{1}{2} \\ t = \frac{1}{2} \\ \text{True} \end{array}$$



Da ϕ_3 keine Singularität des Cotangens im Träger $[0.1, 0.5]$ hat, integrieren wir einfach mit der Cotangensfunktion statt ihrer Regularisierung und vergleichen:

```
In[61]:= NIntegrate[Cot[t]  $\phi_3[t]$ , {t, 0.1, 0.5}, WorkingPrecision -> 20]
(* Cot angewandt auf  $\phi_3$  *)
```

$$\text{Out[61]= } 1.5586306517684558593$$

```
In[62]= NIntegrate[cot_r[200, t]  $\phi_3$ [t], {t, 0.1, 0.5}, WorkingPrecision  $\rightarrow$  40]
(* hier stattdessen mit einer Partialsumme
unserer verallgemeinerten Fourierreihe integriert *)
```

```
Out[62]= 1.558583583272253059535815489899583571810
```

```
In[63]= integrand[t_] := Sum[Cos[2 k t] / k, {k, 1, 200}] f[t]
NIntegrate[integrand[t], {t, 0.1, 0.5}, WorkingPrecision  $\rightarrow$  40]
(* hier stattdessen mittels partieller
Integration die Fourierreihenentwicklung von T benutzt *)
```

```
Out[64]= 1.558583583272253059535815489899583571808
```

```
In[65]= integrand[t_] := -Log[2 Sin[t]] f[t] (* und hier noch  $\leftarrow$ -T,
 $\phi_3'$  > statt wie gerade die Fourierreihennäherung zu benutzen *)
```

```
In[66]= NIntegrate[integrand[t], {t, 0, 0.5}, WorkingPrecision  $\rightarrow$  20]
```

```
Out[66]= 1.5586306517684558564
```

3.3 Zu Distributionen bei mehrdimensionalem Parameterraum

Alle besprochenen Begriffe aus der Distributionentheorie lassen sich übertragen, wenn der zugrunde liegende Parameterraum M mehrdimensional ist. So sind z.B. alle über \mathbb{R}^3 lokal-integrierbaren Funktionen, alle L^1 - oder L^2 -Funktionen f über \mathbb{R}^3 auch als Distributionen aus D' oder auch S' via $\langle f, \phi \rangle = \int f \phi d\lambda^3$ aufzufassen (λ^3 das Lebesgue-Maß auf \mathbb{R}^3).

Die Dirac-Distribution $K \delta(x - x_0)$ für $x, x_0 \in \mathbb{R}^3$ mit $\langle K \delta(x - x_0), \phi(x) \rangle = K \phi(x_0)$ ist ein Modell für eine Punktladung K oder eine Punktmasse K an der Stelle x_0 .

Für den 3-dimensionalen Parameterraum \mathbb{R}^3 als Beispiel werden die **partiellen verallgemeinerte Ableitungen** $\partial^k T = \partial_1^{k_1} \dots \partial_3^{k_3} T$ für einen Multiindex ($\partial_m^n = \partial^n / \partial x_m^n$ n-fache partielle Ableitung nach der m-ten Komponente von $x \in \mathbb{R}^3$)

$$k = (k_1, k_2, k_3), \quad |k| = k_1 + k_2 + k_3, \quad k_i \in \mathbb{N}_0,$$

mit $\phi \in D(\mathbb{R}^3)$ analog zum 1-dimensionalen Fall definiert durch

$$\langle \partial^k T, \phi \rangle = (-1)^{|k|} \langle T, \partial^k \phi \rangle.$$

Hier seien nur drei kurze Beispiele genannt, welche die Nützlichkeit etwa in der theoretischen Physik andeuten mögen:

Beispiel 1. Die Funktion $Q/|x| = Q(x_1^2 + x_2^2 + x_3^2)^{-1/2}$ für $x = (x_1, x_2, x_3) \in \mathbb{R}^3$, $Q > 0$, ist lokal-integrierbar, also eine **reguläre Distribution**. Sie stellt das im Unendlichen verschwindende Potential einer Punktladung Q im Nullpunkt dar.

Beispiel 2. Die **singuläre Distribution** $\rho(x) = \sum_{k=1}^n q_k \delta(x - x_k)$ stellt eine

(verallgemeinerte) Ladungsdichte für n elektrische Punktladungen q_1, \dots, q_n oder eine Massendichte für n elektrische Punktmassen q_1, \dots, q_n an den Raumstellen x_1, \dots, x_n dar.

Beispiel 3. Die **singuläre Distribution** $T = \rho(x) \delta(|x|-R)$, $R > 0$, mit $\phi \in D'(\mathbb{R}^3)$ definiert durch das Oberflächenintegral über die Kugeloberfläche mit Radius R um den Nullpunkt ($d\sigma$ das Oberflächenmass),

$$\langle T, \phi \rangle = \int_{|x|=R} \rho(x) \phi(x) d\sigma(x),$$

ist ein mathematisches Modell für eine **einfache Belegung der Kugeloberfläche mit einer Flächenladungsdichte ρ** . Die Distribution $\delta(|x|-R)$ spielt auch eine wichtige Rolle bei der Lösung der **3D-Wellengleichung** (vgl. [1], Abschnitt 11.6).

Da dieses Notebook nur eine erste Einführung in die Thematik sein soll, gehe ich hier nicht weiter auf den mehrdimensionalen Fall ein. Selbstverständlich spielen Distributionen über \mathbb{R}^3 eine wichtige Rolle in der Theorie der linearen partiellen Differentialgleichungen und haben dort wie auch bei gewöhnlichen Differentialgleichungen eine Erweiterung des Lösungsbegriffs ermöglicht, wenn man diese Gleichungen als solche zwischen Distributionen auffasst. Mehr dazu aber erst später einmal in einem anderen Notebook über Anwendungen der Distributionentheorie. Ich definiere nur schon gleich hier in den beiden letzten Abschnitten nützliche Begriffe zur Koordinatentransformation und zu Faltungen von Distributionen so, dass sie schon für allgemein n -dimensionale Parameterräume gelten.

3.4 Zu Koordinatentransformationen bei Distributionen

Für Distributionen T , eine C^∞ -Transformation $A: \mathbb{R}^n \rightarrow \mathbb{R}^n$ mit Umkehrabbildung A^{-1} und deren Funktionaldeterminante $\partial A^{-1} \neq 0$ sowie eine Testfunktion ϕ aus D definiert man wie beim Transformationssatz für Integrale regulärer Distributionen $T=f$ die Variablensubstitution durch

$$\text{(\#\#)} \quad \langle T(A(x)), \phi(x) \rangle = \langle T(x), |\partial A^{-1}| \phi(A^{-1}(x)) \rangle \quad (x \in \mathbb{R}^n).$$

Damit gelingt es z.B. Translationen und Rotationen von Distributionen einzuführen oder zu definieren, was mit *periodischen, geraden oder ungeraden Distributionen* gemeint ist.

Beispiel 1. Für $A(t)=at+b$, $a \neq 0$, ist $\langle \delta(at+b), \phi(t) \rangle = \langle \delta(t), |a|^{-1} \phi(\frac{t-b}{a}) \rangle$,

$$\text{also } \delta(at+b) = |a|^{-1} \delta(t + \frac{b}{a}).$$

Demo-Beispiel: Es genügt über den Träger von $\delta(at+b)$, nachfolgend ist das die Punktmenge $\{-3/2\}$, zu "integrieren". **Ausserdem ist $\delta(t)=\delta(-t)$, d.h. δ ist eine gerade Distribution.**

```
In[67]:= Integrate[DiracDelta[2 t + 3] gauss[t], {t, -2, -1}]
          gauss[-3/2] / 2
```

```
Out[67]= 1 / (2 e^(9/8))
```

```
Out[68]= 1 / (2 e^(9/8))
```

```
In[69]:= DiracDelta[-t]
```

```
Out[69]= DiracDelta[t]
```

Beispiel 2. Eine Distribution T ist p -periodisch, wenn $\langle T(t+kp), \phi(t) \rangle = \langle T(t), \phi(t-kp) \rangle$ für alle k aus \mathbb{Z} , ϕ aus D oder S gilt. Konkrete Beispiele einer periodischer Distributionen hatten wir 3.2. oben mit der π -periodischen verallgemeinerten Fourierreihe von \cot , und mit der aus den Dirichletkernen entstehenden 2π -periodischen periodischen Impulsfolge schon kennen gelernt.

Beispiel 3. Die Distribution $|x - x_0|^{-1}$ ($x \in \mathbb{R}^3$) ist eine Translation der (regulären) Distribution $|x|^{-1}$. Sie tritt z.B. in der Coulombformel als Faltungskern auf und wird dort mit einer räumlichen Ladungsdichte gefaltet.

3.5 Zur Faltung von Funktionen und Distributionen

Wie schon bei den Definitionen der verallgemeinerten Ableitung oder bei Koordinatentransformationen geht auch bei der Faltung zunächst von der Faltung klassischer Funktionen aus und verwendet dies für eine verallgemeinernde Definition bei Distributionen. Für integrierbare Funktionen f und g auf \mathbb{R}^n ist die Faltung $f * g$ definiert durch

$$(f * g)(x) = \int f(x-y) g(y) d\lambda^n(y),$$

wobei λ^n das n -dimensionale Lebesgue-Maß bezeichnet. Mit ϕ aus D folgt mit der Transformationsregel für Integrale in anderer Schreibweise

$$\langle f * g, \phi \rangle = \langle f(x), \langle g(y), \phi(x+y) \rangle \rangle.$$

Für Distributionen F und G statt f und g definiert man die Faltung ganz genauso und muss dabei nur beachten, dass man Zusatzbedingungen benötigt, die sicherstellen, dass $\langle g(y), \phi(x+y) \rangle$ wieder eine Testfunktion ist, da der Träger von $\phi(x+y)$ i.A. unbeschränkt ist. Wenn die Faltung $F * G$ existiert, ist sie wieder eine Distribution und es gilt $F * G = G * F$:

$$\langle F * G, \phi \rangle = \langle F(x), \langle G(y), \phi(x+y) \rangle \rangle.$$

Zwei Bedingungen, deren jede die Existenz der Faltung zweier Distributionen F und G sicherstellt, sind

a) **F oder G hat kompakten Träger**

b) **Die Träger von F und G liegen in einem Quadranten Q_+^c oder Q_-^c , mit geeignetem $c \in \mathbb{R}$ von der Form $Q_+^c = \{ (x_1, \dots, x_n) : \text{alle } x_i \geq c \}$ bzw. $Q_-^c = \{ (x_1, \dots, x_n) : \text{alle } x_i \leq c \}$.**

(vgl. [1], Abschnitt 7.7 oder entsprechende Stellen in [2]-[6]).

Für 1-dimensionalen Parameterraum sind also zwei Distributionen F und G faltbar, wenn F und G ihre Träger in einer gemeinsamen Halbachse, z.B. in \mathbb{R}_+ haben. Auf weitere hinreichende Bedingungen zur Existenz von *Faltungen zwischen temperierten Distributionen* komme ich in einem späteren Notebook über Fouriertransformationen von Distributionen aus S' und den *Faltungssatz für die Fouriertransformation* noch zu sprechen.

(1) *Faltungen sind kommutativ, distributiv und assoziativ, sofern die auftretenden Faltungen möglich sind.*

Die Dirac-Distribution ist neutral bzgl. Faltung:

Die besondere Rolle der Dirac-Distribution δ besteht darin, dass sie neutral bzgl. Faltung ist; sie ist Faltungs-Eins, d.h. **sie ist Eins-Element in der Faltungsalgebra $D_+' =$** Menge aller Distributionen mit Träger in $[0, \infty[$.

(2) **Für alle Distributionen T gilt $T * \delta = T$.**

Mathematica bietet für die Faltung den Befehl **Convolve** und kennt natürlich diese Eigenschaft, hier als Beispiel ganz allgemein:

In[70]=

```
Convolve[DiracDelta[x], function[x], x, t]
```

Out[70]= function[t]

Fundamentale Eigenschaften von Faltungen sind weiterhin:

(3) *Faltungen sind beliebig mit verallgemeinerten Ableitungen vertauschbar, z.B. also*

$$\partial_{n+m}(T * G) = \partial_n T * \partial_m G = T * \partial_{n+m} G \quad \text{für beliebige natürliche } n, m$$

(4) *Faltungen sind mit Translationen vertauschbar, z.B. also*

$$T(t+p) * G(t) = T * G(t+p) = (T * G)(t+p)$$

Nun ein paar Beispiele mit *Mathematica*.

Beispiel 1. Faltung von zwei Gauss-Kernen ergibt wieder einen Gauss-Kern:

$$\text{In[71]:= } g[\mathbf{x}_-, \mathbf{s}_-, \mathbf{m}_-] = 1 / \left(\mathbf{s} \sqrt{2 \pi} \right) e^{-\frac{(\mathbf{x}-\mathbf{m})^2}{2 \mathbf{s}^2}}$$

$$\text{Out[71]= } \frac{e^{-\frac{(-\mathbf{m}+\mathbf{x})^2}{2 \mathbf{s}^2}}}{\sqrt{2 \pi} \mathbf{s}}$$

$$\text{In[72]:= } \text{Convolve}[g[\mathbf{x}, \mathbf{s}_1, \mathbf{m}_1], g[\mathbf{x}, \mathbf{s}_2, \mathbf{m}_2], \mathbf{x}, \mathbf{t}]$$

$$\text{Out[72]= } \frac{e^{-\frac{(-\mathbf{t}+\mathbf{m}_1+\mathbf{m}_2)^2}{2 (\mathbf{s}_1^2+\mathbf{s}_2^2)}}}{\sqrt{2 \pi} \mathbf{s}_1 \sqrt{\frac{1}{\mathbf{s}_1^2} + \frac{1}{\mathbf{s}_2^2}} \mathbf{s}_2}$$

Ein Beispiel zu (3) und (4) zuerst per Hand und dann mit *Mathematica*:

Beispiel 2. Zur Vertauschbarkeit von Faltung, Ableitung und Translationen

Als Beispiel falten wir die zweite verallgemeinerte Ableitung der a verschobenen Heaviside-Funktion mit einer Distribution F :

$$\begin{aligned} F(t) * \partial_2 \text{HeavisideTheta}(t+a) &= F'(t+a) * \text{HeavisideTheta}'(t) \\ &= F'(t+a) * \delta(t) = F'(t+a). \end{aligned}$$

$$\text{In[73]:= } \text{Convolve}[F[\mathbf{x}], \text{D}[\text{HeavisideTheta}[\mathbf{x} + \mathbf{a}], \{\mathbf{x}, 2\}], \mathbf{x}, \mathbf{t}]$$

$$\text{Out[73]= } F'[\mathbf{a} + \mathbf{t}]$$

Beispiel 3. Faltung der Distributionen $\cos(t)$ HeavisideTheta(t) und

$\sin(t)$ HeavisideTheta(t) mit Träger in \mathbb{R}_+ . Das Ergebnis zeigt z.B., dass ein lineares Übertragungssystem mit rationaler Übertragungsfunktion instabil ist, sobald die Übertragungsfunktion Polstellen hat, die auf der imaginären Achse liegen. Wären etwa i und $-i$ einfache Pole, dann träte ein \cos -Term wie unten in der Impulsantwort auf. Input $\sin(t)$ ab $t=0$ ergäbe dann das nachfolgende mit t unbeschränkt wachsende Ergebnis (vgl. [1], Kap. 10 oder unter dieser URL noch nachfolgende Notebooks über lineare Übertragungssysteme als Anwendung der Distributionentheorie).

Beispiel 4. Diskrete Faltung zweier summierbarer Impulsfolgen

Faltungen von Impulsfolgen spielen in der diskreten Signalverarbeitung eine fundamentale Rolle. Bedingungen, für welche Impulsfolgen Faltungen möglich sind, finden Sie in [1], Kap. 9. Wir betrachten hier die Faltung zweier absolut-summierbarer Impulsfolgen der Form $\sum_{k=0}^{\infty} A_k \delta(t-ka)$ und $\sum_{k=0}^{\infty} B_k \delta(t-ka)$ mit $a>0$. Das Faltungsergebnis ist dann $\sum_{k=0}^{\infty} C_k \delta(t-ka)$ mit

$$C_k = \sum_{m=0}^{\infty} A_m B_{k-m} \quad \text{mit } B_n = 0 \text{ für } n < 0.$$

Zur Berechnung der Faltung genügt es mit den Koeffizienten zu rechnen. *Mathematica* stellt für diskrete Faltungen den Befehl **DiscreteConvolve** bereit.

Hier ein Beispiel mit zwei konkret gegebenen Folgen:

```
In[74]:= f1[n_] = 1 / 2^n UnitStep[n]
          f2[n_] = 1 / 3^n UnitStep[n]
```


```
Out[74]= 2^-n UnitStep[n]
```

```
Out[75]= 3^-n UnitStep[n]
```

```
In[76]:= DiscreteConvolve[f1[n], f2[n], n, k, Assumptions -> k > 0]
```

```
Out[76]= 3 x 2^-k - 2 x 3^-k
```

```
In[77]:= DiscretePlot[3 x 2^-k - 2 x 3^-k, {k, 1, 20}, PlotRange -> All]
```



Wir werden diskrete Faltungen in einem anderen Notebook unter dieser URL in Anwendungsbeispielen zur Signalverarbeitung mit diskreten linearen Filtern wieder begegnen. Die dort auftretenden linearen Differenzgleichungen können als diskrete Faltungen gelesen werden.

Zur Konvergenz von Faltungen

Gilt $T = D' - \lim_{n \rightarrow \infty} T_n$ und ist G eine Distribution aus D' , dann gilt

$$T * G = D' - \lim_{n \rightarrow \infty} (T_n * G)$$

falls eine der folgenden Bedingungen erfüllt ist:

- alle Träger der T_n liegen in einer beschränkten Menge
- G hat einen kompakten Träger
- die Träger aller T_n und von G liegen in einem gemeinsamen Quadranten Q_+^c oder Q_-^c (zur Definition der Quadranten siehe weiter oben).


Beispiel 5. Zur Demonstration mit *Mathematica* betrachten als einfaches Beispiel die Rechteckfunktionen $r[n,t] = \text{UnitStep}[t] - \text{UnitStep}[t-n]$ mit $D' - \lim_{n \rightarrow \infty} r[n,t] = \text{UnitStep}[t]$, bilden die Faltungen $r[n,t] * \text{UnitStep}[t]$ und den D' -lim dieser Faltungen, der dann natürlich $(\text{UnitStep} * \text{UnitStep})[t] = t \text{UnitStep}[t]$ ergibt.

```
In[78]:= r[n_, t_] := UnitStep[t] - UnitStep[t - n]
         Convolve[UnitStep[s], r[5, s], s, t]
```

```
Out[79]= -(-5 + t) UnitStep[-5 + t] + t UnitStep[t]
```

```
In[80]:= Plot[%, {t, -2, 10}, PlotStyle -> Directive[
         Blue, Thickness[0.006]]]
```

```
Out[80]=
```




```
In[81]:= Limit[Convolve[UnitStep[s], r[n, s], s, t], n -> ∞]
```

```
Out[81]= { t t ≥ 0
          { 0 True
```

```
In[82]:= Plot[%, {t, -1, 10}, PlotStyle -> Directive[
         Blue, Thickness[0.006]]]
```

```
Out[82]=
```



Dies sei als Einstieg in etwas Distributionentheorie mit ein paar Demonstrationen, wie man *Mathematica* verwenden kann, um mit dem Distributionenkalkül konkrete Beispiele zu rechnen, hier genug. Insbesondere die Berechnung von Faltungen wird i.A. schnell ein Problem sein, wenn die beteiligten Funktionen oder Distributionen nicht mehr so simpel wie in unseren Demo-Beispielen sind. Für gewisse Klassen von Funktionen können Transformationen wie die *Fouriertransformation*, die *Laplace-Transformation* oder die *Mellin-Transformation* mit ihren Faltungssätzen sehr hilfreich sein. Ich gehe darauf in nachfolgenden Notebooks näher ein.

Literaturhinweise : Bei mehr Interesse an Distributionen empfehle ich neben [1]

- [2] I.M. Gel'fand, G.E. Shilov Generalized Functions, Vol. I, Academic Press 1964
- [3] B. Fuchssteiner, D. Laugwitz Funktionalanalysis, BI Wissenschaftsverlag 1974
- [4] V.S. Vladimirov Methods of the Theory of Generalized Functions,
Taylor and Francis 2002
- [5] W. Walter Einführung in die Theorie der Distributionen,
BI Wissenschaftsverlag 1994
- [6] A.H. Zemanian Distribution Theory And Transform Analysis,
Dover 1965
- [7] E. Novak, M. Ullrich,
H. Wozniakowsky Complexity of Oscillatory Integration for Univariate
Sobolev Spaces, arXiv: 1311. 1528v2 [math.NA], 2014,
J. Complexity, to appear